
HP-33S Calculator Program Curves 1
Compute Values for a Parabolic Vertical Curve

Programmer: Dr. Bill Hazelton

Date: March, 2008. Version: 1.0 Mnemonic:  P for Parabolic Vertical Curve.

Line Instruction Display User Instructions
P0001 LBL P   LBL  P
P0002 CLx   CLEAR  1
P0003 ENTER
P0004 ENTER
P0005 ENTER
P0006 SF  10   FLAGS  1  .0
P0007 COMP  V  CURVE   EQN RCL C, RCL O,  etc.
P0008 PSE   PSE
P0009 CLx   CLEAR  1
P0010 STO  S
P0011 STO  R
P0012 STO  Q
P0013 STO  P
P0014 STO  A
P0015 STO  L
P0016 ENTER  P  I  RD   EQN RCL E, RCL N,  etc.
P0017 PSE   PSE
P0018 INPUT  R R?   INPUT  R
P0019 ENTER  P  I  EL   EQN RCL E, RCL N,  etc.
P0020 PSE   PSE
P0021 INPUT  S S?   INPUT  S
P0022 START GRADE   EQN RCL S, RCL T,  etc.
P0023 PSE   PSE
P0024 INPUT  P P?   INPUT  P
P0025 END GRADE   EQN RCL E, RCL N,  etc.
P0026 PSE   PSE
P0027 INPUT  Q Q?   INPUT  Q
P0028 RCL  Q
P0029 RCL—  P
P0030 STO  A
P0031 ENTER  LENGTH   EQN RCL E, RCL N,  etc.
P0032 PSE   PSE
P0033 INPUT  L L?   INPUT  L
P0034 RCL  R
P0035 RCL  L
P0036 2
P0037 ÷
P0038 —
P0039 STO  U
P0040 RCL  S

Line Instruction Line Instruction Line Instruction
P0041 RCL  P P0081 VIEW  U G0012 +
P0042 RCL×  L P0082 START  PT  EL G0013 RCL+  V
P0043 2 P0083 PSE G0014 STO  H
P0044 ÷ P0084 VIEW  V G0015 EL  OF  POINT
P0045 — P0085 END  PT  RD G0016 PSE
P0046 STO  V P0086 PSE G0017 VIEW  H
P0047 RCL  A P0087 RCL  U G0018 0
P0048 RCL÷  L P0088 RCL+  L G0019 STO  Y
P0049 2 P0089 STO  F G0020 AGAIN  (0–1)
P0050 ÷ P0090 VIEW  F G0021 PSE
P0051 STO  B P0091 RCL  L G0022 INPUT  Y
P0052 RCL  P P0092 x2 G0023 RCL  Y
P0053 RCL÷  B P0093 RCL×  B G0024 x > 0?
P0054 2 P0094 RCL  L G0025 GTO  G
P0055 ÷ P0095 RCL×  P ***** Get RD from EL
P0056 +/— P0096 + H0001 LBL  H
P0057 STO  I P0097 RCL+  V H0002 0
***** Max/Min Pt. Data P0098 STO  E H0003 STO  Y
P0058 MAX-MIN VALUES P0099 END  PT  EL H0004 STO  H
P0059 PSE P0100 PSE H0005 STO  X
P0060 RCL  I P0101 VIEW  E H0006 COMP  1  RD (0–1)
P0061 x2 ***** Get EL from RD H0007 PSE
P0062 RCL×  B P0102 0 H0008 INPUT  Y
P0063 RCL  I P0103 STO  Y H0009 RCL  Y
P0064 RCL×  P P0104 STO  X H0010 x ≤ 0?
P0065 + P0105 COMP  1  EL (0–1) H0011 GTO  L
P0066 RCL+  V P0106 PSE H0012 ENTER  EL
P0067 STO  E P0107 INPUT  Y H0013 PSE
P0068 MAX—MIN  EL P0108 RCL  Y H0014 INPUT  H
P0069 PSE P0109 x ≤ 0? H0015 RCL  V
P0070 VIEW  E P0110 GTO  H H0016 RCL—  H
P0071 MAX—MIN RD G0001 LBL  G H0017 STO  C
P0072 PSE G0002 ENTER RD H0018 RCL  P
P0073 RCL  U G0003 PSE H0019 x2

P0074 RCL+  I G0004 INPUT  X H0020 RCL  B
P0075 STO  D G0005 RCL  X H0021 RCL×  C
P0076 VIEW  D G0006 RCL—  U H0022 4
***** End Pts Data G0007 STO  I H0023 ×
P0077 END POINTS G0008 x2 H0024 —
P0078 PSE G0009 RCL×  B H0025 STO  D
P0079 START PT RD G0010 RCL  I H0026 x < 0?
P0080 PSE G0011 RCL×  P H0027 THIS  ELEV

*****  These lines are simply comments in the code. You don’t key it into the calculator!

Line Instruction Line Instruction Line Instruction
H0028 x < 0? L0005 STO  H L0049 VIEW  H
H0029 PSE L0006 STO  D L0050 RCL  C
H0030 x < 0? L0007 STO  C L0051 STO  I
H0031 NOT  ON  CURVE L0008 STO  I K0001 LBL  K
H0032 x < 0? L0009 STEP THRU RD K0002 RCL  D
H0033 PSE L0010 PSE K0003 STO+  I
H0034 x < 0? L0011 NO—YES  (0–1) K0004 RCL  I
H0035 GTO  L L0012 PSE K0005 RCL—  L
H0036 RCL  D L0013 INPUT  Y K0006 x ≥ 0?
H0037 √x L0014 RCL  Y K0007 GTO  O
H0038 STO  D L0015 x ≤ 0? K0008 RCL  I
H0039 RCL—  P L0016 CF  10 K0009 x2

H0040 RCL÷  B L0017 x ≤ 0? K0010 RCL×  B
H0041 2 L0018 RTN K0011 RCL  I
H0042 ÷ L0019 FIRST INCRMNT K0012 RCL×  P
H0043 RCL+  U L0020 PSE K0013 +
H0044 STO  X L0021 INPUT  C K0014 RCL+  V
H0045 FIRST  RD L0022 GENRL INCRMNT K0015 STO  H
H0046 PSE L0023 PSE K0016 RCL  I
H0047 VIEW  X L0024 INPUT  D K0017 RCL+  U
H0048 RCL  D L0025 FIRST  PT K0018 STO  X
H0049 +/— L0026 PSE K0019 RD VALUE
H0050 RCL—  P L0027 RD  VALUE K0020 PSE
H0051 RCL÷  B L0028 PSE K0021 VIEW  X
H0052 2 L0029 VIEW  U K0022 EL  VALUE
H0053 ÷ L0030 EL  VALUE K0023 PSE
H0054 RCL+  U L0031 PSE K0024 VIEW  H
H0055 STO  X L0032 VIEW  V K0025 GTO  K
H0056 SECOND  RD L0033 RD VALUE O0001 LBL  O
H0057 PSE L0034 PSE O0002 END POINT
H0058 VIEW  X L0035 RCL  C O0003 PSE
H0059 0 L0036 RCL+  U O0004 RCL  L
H0060 STO  Y L0037 STO  X O0005 x2

H0061 AGAIN  (0–1) L0038 VIEW  X O0006 RCL×  B
H0062 PSE L0039 RCL  C O0007 RCL  L
H0063 INPUT  Y L0040 x2 O0008 RCL×  P
H0064 RCL  Y L0041 RCL×  B O0009 +
H0065 x > 0? L0042 RCL  C O0010 RCL+  V
H0066 GTO  H L0043 RCL×  P O0011 STO  H
***** Step thru RDs L0044 + O0012 RCL  L
L0001 LBL  L L0045 RCL+  V O0013 RCL+  U
L0002 0 L0046 STO  H O0014 STO  X
L0003 STO  Y L0047 EL  VALUE O0015 RD  VALUE
L0004 STO  X L0048 PSE O0016 PSE

Line Instruction
O0017 VIEW  X
O0018 EL  VALUE
O0019 PSE
O0020 VIEW  H
O0021 CF  10
O0022 RTN

Notes

1. The **** lines are comments and are not to be entered into the calculator. They are there
to make it easier to work through entering a long program.

2. The program offers options for the processing to be done at some steps, asking the user if
a particular set of operations are to be done. The user is prompted for a yes/no answer,
with the input variable being Y, which is set to 0 (no) by default. To skip the operation,
just press R/S. To do the operation, key in 1, then press R/S. The sense is that 0 = no, 1 =
yes. Note that there is a need to pause after the prompt, otherwise the entered value will
not make it into the calculator, and the ‘no’ response will be acted upon.

3. Because grades are used, it is important that the units for elevation and running distance
are the same. Otherwise the computations will not be correct.

4. Some of the calculations will yield results that are on the parabola being used, but not
within the actual segment being used. This is particularly the case with finding maximum
and minimum points, which may not lie within the limits of the curve, and finding running
distances given elevations. The user should check that the running distances fall within the
start and end points of the curve and ignore results that lie outside the curve. The program
does not check these limits.

Theory and Background

The theory of computing the various values for a parabolic ‘equal-tangent-length’ vertical curve
is fairly straightforward. There are six basic possibilities of how the vertical curve can be placed
between two grade lines, as shown in the figure overleaf. The curve allows the grade to be
changed smoothly from the incoming grade (p) to the outgoing grade (q).

The convention with vertical curves is that they are drawn and computed going from left to right.
If the grade is rising from left to right, the gradient is positive. If it is going down from left to
right, the gradient is negative. The diagram overleaf shows the signs that the various gradients
would take in the circumstances.

If the grades on opposite sides of the curve have the same sign, as in the two examples in the right
in the figure above, there is no maximum or minimum value along the curve (other than at its end
points). If the grades have opposite signs (as in the two examples on the left side of the above
diagram), then there will be a maximum or minimum point of the parabola somewhere along the
curve. In this case, the calculation of the elevation and location of such a point is meaningful. If
such a calculation is made for the parabolas on the upper and central right and along the bottom in
the diagram, the turning point for the parabola will be determined, but it will be well outside the

part of the parabolic curve actually used. For example, the parabolas shown on the upper and
central right side of the diagram above have their turning points well to the right of the end of the
curve, while those along the bottom have their turning points well to the left of the actual curve.

+p%

–p% +q%

–q%

Difference in gradients
( q – p )

Difference in gradients
( q – p )

+p%

+q%

–p%

–q%

–q%

+p%

For the purposes of vertical curve design, parabolas are the preferred curve. They are simpler to
compute than circles or ellipses, but differ from them by amounts that are too small to matter in
almost all cases. With a parabola, there is a constant change of grade (or gradient) around the
curve, whereas for a circle there is a constant change of angle around the curve.

Another useful characteristic of the parabola is that it can be placed so that the lengths of the
tangents are always equal. As the tangents are usually close to level (grades usually being fairly
small), the horizontal distances from the point of intersection of the two grade lines to the tangent
points are equal, and half the length of the curve. This allows easy placement of the curve with
respect to the point of intersection of the grade lines. [Note that if the in and out grades are not
equal and of opposite sign, the maximum or minimum point will not fall directly below the
intersection point.]

Gradients, Grades, Slopes and Angles

The slopes of the lines into and out of the vertical curve may be expressed in several ways. For
this program, gradients should be entered as a decimal value of the gradient.

The gradient is the value of the change in elevation over a horizontal distance divided by that
distance, i.e., rise over run, expressed as a decimal number. So, if the slope rises by 2 units for
every 100 along, the gradient is +2 ÷ 100  =  +0.02. If the slope falls by 5 units for 125 units
along, the gradient is  –5 ÷ 125  =  –0.04.

The gradient can also be expressed as a percentage, which is simply the gradient value (as above)
multiplied by 100 to convert it to a percentage. So the above examples would be +2% and –4%,
respectively. Percentages are also handy in that they link in well to horizontal distances expressed
in stations. As the distance is then in 100 ft units, a 1 foot rise would be a +1% grade, so the rise
or fall over one 100 ft ‘station’ can be converted directly to the percentage gradient.

Gradients can also be expressed as a ratio of the rise to the run, and expressed in the form “1 in so
many.” To get this “so many” value, simply calculate the reciprocal value of the gradient (as a
decimal), so for the two example given above, +0.02 would be 1⁄0.02  =  50, and so +1 in 50; –0.04
would be 1⁄0.04  =  25, and so –1 in 25.

Gradients can also be represented as the angle of the line from the horizontal, usually given in
decimal degrees. The tangent of this angle will be the gradient, and the gradient can be converted
to an angle by taking the arctangent of the gradient. So a gradient of +0.02 will give an angle of
arctan(+0.02) = +1.146°, while –0.04 will give and angle of arctan(–0.04) = –2.291°. A slope of
1° would give a gradient of tan(1°) = 0.017.

Grade and gradient are used interchangeably, although sometimes they are applied to specific
representations. For this program, convert all gradients to the decimal format, e.g., +0.02, –0.04.
Be aware that the sign of the gradient is very important and must be included.

The term ‘slope’ is also used, but it usually doesn’t refer to a specific representation.

Horizontal Distances

Horizontal distances as used in the construction of linear objects are commonly expressed as a
distance from a starting point somewhere along the object. How they are expressed depends upon
the units being used, the country in which they are being used, and local practice. Similarly, what
they are called also varies.

Distances in feet are commonly recorded as ‘station’ values. Here, it is assumed that a station is
marked every 100 ft, and that the stations are numbered sequentially from the start, with distances
on from the station noted as additional distance. So a distance of 12,546.78 ft would be recorded
as 125 + 46.78, meaning 125 stations of 100 feet, plus 46.78 feet past that station. For many
construction projects, having points every 100 ft (30.48 m) is very convenient, hence the
popularity of this representation. It is easy to convert between the distance representation
(12,546.78 ft) and the station representation (125 + 46.78): simply remove the + sign and place
the digits together to go to distance, or open the digits two left of the decimal point and put in a +,
to convert to stations. (Calculators prefer the distance version.)

With metric units, 100 m is a bit long for station placement, so an equivalent metric
representation never really caught on. In metric, it is more usual to use the distance representation
(in meters) for all uses. This is also simpler as surveying moves away from the reliance on short
lines for set-out (greatly aided by a station every 100 ft), to total station based set-out by
co-ordinates across large areas.

The distances are known as ‘stations’ (when working with the 100 ft units), but this is a little odd
when using a distance representation. In this case, the distance may be known as the ‘Running
Distance’ (abbreviated RD), or the ‘chainage’ in some circumstances. As the station
representation is so easily converted to the distance representation, and this program can also be
used for metric applications, the horizontal distances in this program will be termed ‘Running
Distances’ and often noted as RD.

Calculating The Parabola

The general equation of a parabola is:

y  =  ax2  +  bx  +  c

The magnitude of the term a controls the sharpness of the parabola, while the sign of a controls
the orientation. With a positive, the parabola is turned upwards and is bowl-shaped (the apex or
turning point is the smallest y value), while with a negative, the parabola is hill-shaped, with the
apex having the largest y value. So a summit or crest has a negative a, while a sag has a positive
a.

It is convenient to use the starting point of the parabola (i.e., the first or left-most tangent point)
as the origin of the co-ordinates. The elevation of this point (A on the diagram overleaf) above the

chosen datum is equal to the term c in the equation. The slope  

� 

dy
dx

 of any tangent is equal to

2ax + b. But as at Point A, x = 0, the term b in the general equation is the slope or gradient at
point A, the tangent gradient p.

The second derivative 

� 

d 2y
dx 2

 of the general parabola equation equals 2a, a constant. This means

that a tangent to a vertical-axis parabola changes a constant amount of grade for each increment
of distance. (In contrast, a tangent to a vertical circular curve changes direction a constant amount
of angle for equal increments of distance along the arc.) The useful consequence is that the rate of
change of grade on a vertical curve is constant and equals 2a per 100 units of distance (feet or
meters, depending upon the units chosen). On a vertical curve the total change in direction
between the profile grades is  q – p, termed A.

If this change is accomplished on a curve L units long, the constant rate of change must be:

2a  =  

� 

q − p
L

  =  

� 

A
L

with gradients q, p and A in decimal form, and L in either feet or meters, consistent with the job.

A practical formula for a general parabola for a vertical curve is therefore

Elevation  =  a (RD – RDA)2  +  p (RD  –  RDA)  +  ElevationA

where RD is the running distance of any point, RDA is the running distance of point A, a is the
first parameter of the parabola (half the rate of change of grade as a decimal), p is the slope of the
entry (or incoming) tangent (in decimal form), Elevation is the elevation of the point at running
distance RD, and ElevationA is the Elevation of point A. Elevation and RD should be in the same
units, consistent with the entire job.

The location of the turning point of the parabola (the apex) can be computed by noting that the
slope of the tangent at the turning point is 0, and solving 2ax + b = 0. In this case, the location of

the turning point is: x = 

� 

−b
2a

. This value can be converted to a running distance by adding the

value of RDA, and then used in the equation to compute the elevation of the maximum or
minimum point on the curve.

Given the Elevation of a point on the curve, its location, x, from Point A can be computed by
solving the equation:

a x2  +  p x  +  (ElevationA – Elevation)  =  0

As this is a quadratic, the standard quadratic solution will produce two solutions (in most cases),
and it is up to the user to decide which is the most applicable. The program deals with the various
cases (0, 1 and 2 solutions) separately. If there is a single solution, the apex (or nadir) of the curve
has been selected. If there are no solutions, then the elevation selected is beyond the turning point
elevation of the curve and therefore cannot lie on the curve.

For determining the elevations at set distances along the curve, the user can specific an initial
increment, to bring the steps onto an even running distance, then specify a general increment,
which will be used for the remainder of the curve. The program calculates the elevation of the
first point on the curve (the left-hand tangent point), then moves along the initial increment, then
proceeds along the curve using steps of the general increment, until the end of the curve is
reached. The final point on the curve (the right-hand tangent point) is calculated and the program
ends. For each point, the running distance and elevation are shown.

Note that if the final point calculated by increments happens to also be the end point, it will be
calculated and displayed, and then the end point will be calculated again and displayed. This is
because the test for coming to the end of the curve is that the increment is beyond the end point.

Running the Program

Key in XEQ  P. The program starts and displays:

COMP  V  CURVE

then prompts for the running distance of the intersection point of the two grade lines, displaying:

ENTER  P  I  RD

then stops while displaying:

R?
0.0000

Key in the RD of the intersection point and press R/S. The calculator then prompts for the
elevation of the intersection point of the two grade lines, displaying:

ENTER  P  I  EL

then stops while displaying:

S?
0.0000

Key in the elevation of the intersection point, then press R/S. The calculator then prompts for the
starting grade (p), displaying:

START  GRADE

then stops while displaying:

P?
0.0000

The value of the slope coming in to the vertical curve should be entered as a decimal value, e.g.,
0.025 (an upward 2.5% gradient), then press R/S. The calculator then prompts for the grade
coming out of the vertical curve, displaying:

END  GRADE

then stops while displaying:

Q?
0.0000

Key in the value of the slope of the grade coming out of the curve, as a decimal grade, e.g., –0.04
(a downward 4% gradient), then press R/S. The calculator then prompts for the length of the
curve, briefly displaying:

ENTER  LENGTH

then stops while displaying:

L?
0.0000

Key in the length of the curve desired and press R/S.

The calculator displays  RUNNING  briefly, and briefly displays:

MAX–MIN  VALUES

then briefly displays:

MAX–MIN  EL

then stops and displays the Elevation of the maximum or minimum point, looking as follows:

E=
      114.5611

Press R/S to continue. The calculator then briefly displays:

MAX–MIN  RD

then stops and displays the Running Distance of the maximum or minimum point, looking as
follows:

D=
      24,589.2222

Press R/S to continue.

This option computes the running distance and elevation of the end points of the curve, where the
parabolic curve joins the tangent gradients. The calculator briefly displays:

END  POINTS

then briefly displays:

START  PT  RD

then stops and display the running distance of the start point of the curve, such as follows:

U=
       24,367.0000

Press R/S to continue. The calculator briefly displays:

START  PT  EL

then stops and displays the elevation of the start point of the curve, such as follows:

V=
         103.4500

Press R/S to continue. The calculator briefly displays:

END  PT  RD

then stops and display the running distance of the end point of the curve, such as follows:

F=
       24,767.0000

Press R/S to continue. The calculator briefly displays:

END  PT  EL

then stops and displays the elevation of the end point of the curve, such as follows:

E=
            107.4500

Press R/S to continue.

A. Compute Elevation at a Specified Running Distance

With this option, the user can enter any running distance and the calculator will compute the
elevation on the curve at that point. The calculator does not check if the running distance is on the
curve segment actually being used, so the user must check this. The calculator briefly displays:

COMP  1  EL  (0–1)

although the right-hand parenthesis will be off screen. The calculator then stops and displays:

Y?
0.0000

If you don’t want to run this option, press R/S and the program will advance to the next option. If
you do want to run this option, key in 1 (or any number greater than 0), which signifies ‘yes,’ and
press R/S. The calculator briefly displays:

ENTER  RD

then stops and prompts for the RD to be entered, displaying:

X?
0.0000

Key in the running distance of the point of interest and press R/S. The calculator then briefly
displays:

EL  OF  POINT

then stops and displays the elevation value of the selected point, such as follows:

H=
           114.3690

Press R/S to continue, and the calculator prompts to see if you want to compute another elevation,
briefly displaying:

AGAIN  (0–1)

then stopping and displaying:

Y?
0.0000

If you want to do another point, key in 1 and press R/S. The program then prompts for the
running distance (as above) and loops through the option until you decide not to do it again. If
you don’t want to do this option, press R/S, and the option ends.

B. Compute the Running Distances at a Specified Elevation

This option allows the user to enter an elevation and compute the running distance(s) at which it
occurs. Since the curve is a parabola, there will be zero, one or two solutions. If the elevation
cannot occur on the curve, i.e., the elevation is beyond the elevation of the turning point of the
parabola, there will be zero answers. If the elevation chosen is that of the turning point, there will
be just one answer, which will be given twice. The other results have two answers.

The calculator briefly displays:

COMP  1  RD  (0–1)

with the right-hand parenthesis off the screen. The calculator then stops and displays the prompt:

Y?
0.0000

If you want to skip this option, just press R/S and the calculator moves on to the next option. If
you do want to run this option, key in 1 and press R/S. The calculator briefly displays:

ENTER  EL

then stops and displays:

H?
0.0000

Key in the elevation of interest, and press R/S.



HP-33S Calculator Program Curves 1
Compute Values for a Parabolic Vertical Curve

- 13 -

If the elevation is not on the curve, the calculator displays:

THIS  ELEV

press R/S to continue, and the calculator displays:

NOT  ON  CURVE

Press R/S again, and you are taken to the next option.

If the elevation is elsewhere on the curve, the calculator briefly displays:

FIRST  RD

then stops and displays the running distance of the first solution point, such as follows:

X=
      24,383.0819

Press R/S and the calculator briefly displays:

SECOND  RD

then stops and displays the running distance of the second solution point, such as follows:

X=
       24,795.3625

It is up to the user to decide if the points fall within the end points of the curve, and chose points
that are useful.

Press R/S to continue, and the calculator then briefly displays:

AGAIN  (0–1)

then stops and displays:

Y?
0.0000

To run the option again, key in and 1 and press R/S. The calculator will then return to the
beginning of the options and prompt to see if you want to run the option, displaying briefly:

COMP  1  RD  (0–1)

with the right-hand parenthesis off the screen. The calculator then stops and displays the prompt:

Y?
0.0000



HP-33S Calculator Program Curves 1
Compute Values for a Parabolic Vertical Curve

- 14 -

Key in 1 to continue with the option. (Yes, I could have cut this out, but it would have taken
another label, and there are already 6 in this program! If you add the issue of fixing the problem
of an elevation not on the curve, there is another label!)

The calculator then prompts for the elevation to be entered, as above, and run through the option
again. If you don’t want to run the option again, press R/S and the option ends.

C. Step Through a Series of Running Distances to get Elevations at each

The final option allows the user to step through a series of equally-spaced points along the curve.
As the start point is often at an odd running distance, this option allows the user to select a first
increment, to allow the running distances to be brought to even values (e.g., exactly onto 100 ft
stations), and then select a general increment to be applied successively until the end point is
reached. The end point is calculated as the last point along the curve.

The option begins by displaying briefly:

STEP  THRU  RD

then displaying briefly:

NO–YES  (0–1)

then stopping and displaying:

Y?
0.0000

If you want to run this option, key in 1 and press R/S. If not, just press R/S and the program ends.
If you are running the option, the calculator briefly displays:

FIRST  INCRMNT

then stops and displays:

C=
0.0000

Key in the first increment, then press R/S. The calculator briefly displays:

GENRL  INCRMNT

then stops and displays:

D=
0.0000

Key in the increment to be used for all the other distances, then press R/S. The calculator then
briefly displays:

FIRST PT

RD  VALUE

then stops and displays the running distance of the start point, such as follows:

U=
      24,367.0000

Press R/S to continue. The calculator briefly displays:

EL  VALUE

then stops and displays the elevation at the start point, such as follows:

V=
        103.4500

Press R/S to continue. The calculator then loops through the following sequence, briefly
displaying:

RD  VALUE

then stopping and displaying the next running distance value, such as follows:

X=
     24,467.0000

Press R/S to continue. The calculator briefly displays:

EL VALUE

then stops and displays the elevation value at the running distance just given, such as follows:

H=
          111.2000

Press R/S to continue through this loop until the increments extend past the last point. At this
stage, the last point is displayed. The calculator briefly displays:

END  POINT

then briefly displays:

RD VALUE

then stops and displays the running distance of the end point, such as follows:

X=
       24,767.0000

Press R/S to continue. The calculator briefly displays:

EL  VALUE

then stops and displays the elevation of the end point, such as follows:

H=
         107.45

Press R/S to continue. The calculator briefly displays

PROGRAM  END

and then stops at the end of the program, having reset Flag 10.

Sample Computations

The sample computations are based on the following general data:

Running Distance of the Point of Intersection  =  24,567.0000
Elevation of the Point of Intersection  =  123.4500
Length of the Curve  =  400.0000

The running distance of the start point is therefore 24,367 and of the end point is 24,767.

The following tabulations show the results for different in and out gradients. The increments used
for the running distance values are both 50 (both the first increment and the general increment).

Case In Grade Out Grade Start EL End EL Max/Min RD Max/Min EL

1 0.0200 -0.0200 119.45 119.45 24567.00 121.45
2 0.0400 -0.0600 115.45 111.45 24527.00 118.65
3 0.0600 -0.0400 111.45 115.45 24607.00 118.65
4 0.0800 -0.1000 107.45 103.45 24544.78 114.56
5 0.1000 -0.0800 103.45 107.45 24589.22 114.56
6 -0.0200 0.0400 127.45 131.45 24500.33 126.12
7 -0.0400 0.0200 131.45 127.45 24633.67 126.12
8 -0.0600 0.0800 135.45 139.45 24538.43 130.31
9 -0.0800 0.1000 139.45 143.45 24544.78 132.34

10 -0.1000 0.0600 143.45 135.45 24617.00 130.95
11 0.0200 0.0400 119.45 131.45 23967.00 115.45
12 0.0400 0.0200 115.45 127.45 25167.00 131.45
13 0.0600 0.0800 111.45 139.45 23167.00 75.45
14 0.0800 0.1000 107.45 143.45 22767.00 43.45
15 0.1000 0.0600 103.45 135.45 25367.00 153.45
16 -0.0200 -0.0400 127.45 115.45 23967.00 131.45
17 -0.0400 -0.0600 131.45 111.45 23567.00 147.45
18 -0.0600 -0.0400 135.45 115.45 25567.00 99.45
19 -0.0800 -0.1000 139.45 103.45 22767.00 203.45
20 -0.1000 -0.0800 143.45 107.45 26367.00 43.45

Running Distance
Case 24367 24417 24467 24517 24567 24617 24667 24717 24767

1 119.45 120.33 120.95 121.33 121.45 121.33 120.95 120.33 119.45
2 115.45 117.14 118.20 118.64 118.45 117.64 116.20 114.14 111.45
3 111.45 114.14 116.20 117.64 118.45 118.64 118.20 117.14 115.45
4 107.45 110.89 113.20 114.39 114.45 113.39 111.20 107.89 103.45
5 103.45 107.89 111.20 113.39 114.45 114.39 113.20 110.89 107.45
6 127.45 126.64 126.20 126.14 126.45 127.14 128.20 129.64 131.45
7 131.45 129.64 128.20 127.14 126.45 126.14 126.20 126.64 127.45
8 135.45 132.89 131.20 130.39 130.45 131.39 133.20 135.89 139.45
9 139.45 136.01 133.70 132.51 132.45 133.51 135.70 139.01 143.45

10 143.45 138.95 135.45 132.95 131.45 130.95 131.45 132.95 135.45
11 119.45 120.51 121.70 123.01 124.45 126.01 127.70 129.51 131.45
12 115.45 117.39 119.20 120.89 122.45 123.89 125.20 126.39 127.45
13 111.45 114.51 117.70 121.01 124.45 128.01 131.70 135.51 139.45
14 107.45 111.51 115.70 120.01 124.45 129.01 133.70 138.51 143.45
15 103.45 108.33 112.95 117.33 121.45 125.33 128.95 132.33 135.45
16 127.45 126.39 125.20 123.89 122.45 120.89 119.20 117.39 115.45
17 131.45 129.39 127.20 124.89 122.45 119.89 117.20 114.39 111.45
18 135.45 132.51 129.70 127.01 124.45 122.01 119.70 117.51 115.45
19 139.45 135.39 131.20 126.89 122.45 117.89 113.20 108.39 103.45
20 143.45 138.51 133.70 129.01 124.45 120.01 115.70 111.51 107.45

These values should allow the program to be tested to make sure it is working properly. The
above tabulated values were calculated by spreadsheet, rather than the calculator, but the
calculator results were checked against these tabulations.

Storage Registers Used

A Difference between the incoming and outgoing gradients.

B Parameter a in the parabola equation.

C Elevation difference. First increment value for ‘stepping’ option.

D Running Distance of a computed point. General increment for ‘stepping’ option.

E Elevation of a computed point.

F Running Distance of a computed point.

H Elevation of a computed point, and entered elevation to have RD calculated.

I Distance along curve from start point.

L Length of the curve to be computed.

P Gradient of the incoming tangent (start grade)

Q Gradient of the outgoing tangent (end grade).

R Running Distance of the Point of Intersection.

S Elevation of the Point of Intersection.

U Running Distance of the start point.

V Elevation of the start point.

X Running Distance entered to compute elevation at that point.

Y Yes/No variable for option choices.

Statistical Registers: Not used.

Labels Used

Label P Length  =  557 Checksum  =  5A1D

Label G Length  =  117 Checksum  =  5CA7

Label H Length  =  330 Checksum  =  2187

Label L Length  =  255 Checksum  =  3C67

Label K Length  =    91 Checksum  =  1C67

Label O Length  =    91 Checksum  =  D6ED

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.

Flags Used

Flag 10 is used by this program. Flag 10 is set for this program, so that equations can be shown as
prompts. At the end of the program, Flag 10 is cleared.


